Fortran pointers

Victor Eijkhout, Susan Lindsey

Fall 2021

last formatted: November 11, 2021

1. Fortran Pointers

- Pointer points at a variable of any type:
  elementary, or derived types.
- You can access a variable through pointer,
  including changing it
- You can change what variable the pointer points at.
- A pointer acts like an alias:
  no explicit dereference needed.

2. Setting the pointer

- You have to declare that a variable is pointable:

```
real,target :: x
```

- Declare a pointer:

```
real,pointer :: point_at_real
```

- Set the pointer with => notation (New! Note!):

```
point_at_real => x
```

3. Dereferencing

Fortran pointers are often automatically *dereferenced*: if you print a pointer you print the variable it references, not some representation of the pointer.

Code:

```
real,target :: x
real,pointer :: point_at_real

x = 1.2
point_at_real => x
print *,point_at_real
```

Output

```
[pointerf] basicp:

   1.20000005
```

4. Pointer example

Code:

```
real,target :: x,y
real,pointer :: that_real

x = 1.2
y = 2.4
that_real => x
print *,that_real
that_real => y
print *,that_real
y = x
print *,that_real
```

Output

```
[pointerf] realp:

   1.20000005
   2.40000010
   1.20000005
```

1. The pointer points at x, so the value of x is printed.
2. The pointer is set to point at y, so its value is printed.
3. The value of y is changed, and since the pointer still points at y, this changed value is printed.

5. Assign pointer from other pointer

```
real,pointer :: point_at_real,also_point
point_at_real => x
also_point => point_at_real
```

Now you have two pointers that point at `x`.

Very important to use the =>, otherwise strange memory errors

6. Assignment subtleties

Assign underlying variables:

```
real,target :: x,y
real,pointer :: p1,p2

x = 1.2
p1 => x
p2 => y
p2 = p1 ! same as y=x
print *,p2 ! same as print y
```

Crash because *p2* pointer unassociated:

```
real,target :: x
real,pointer :: p1,p2

x = 1.2
p1 => x
p2 = p1
print *,p2
```

7. Pointer status

- `Nullify`: zero a pointer
- `Associated`: test whether assigned

Code:

```
real,target :: x
real,pointer :: realp

print *,"Pointer starts as not set"
if (.not.associated(realp)) &
   print *,"Pointer not associated"
x = 1.2
print *,"Set pointer"
realp => x
if (associated(realp)) &
   print *,"Pointer points"
print *,"Unset pointer"
nullify(realp)
if (.not.associated(realp)) &
   print *,"Pointer not associated"
```

Output
[pointerf] statusp:

```
 Pointer starts as
    not set
 Pointer not
    associated
 Set pointer
 Pointer points
 Unset pointer
 Pointer not
    associated
```

8. Pointer allocation

If you want to hang an object from a pointer, but you don't need a variable too:

Code:

```
Real,pointer :: x_ptr,y_ptr
allocate(x_ptr)
y_ptr => x_ptr
x_ptr = 6
print *,y_ptr
```

Output

```
[pointerf] allocptr:

   6.00000000
```

Exercise 1

Write a routine that accepts an array and a pointer, and on return has that pointer pointing at the largest array element:

Code:

```
real,dimension(10),target :: array &
     = [1.1, 2.2, 3.3, 4.4, 5.5, &
        9.9, 8.8, 7.7, 6.6, 0.0]
real,pointer :: biggest_element

print '(10f5.2)',array
call SetPointer(array,biggest_element)
print *,"Biggest element is",
     biggest_element
print *,"checking pointerhood:",&
     associated(biggest_element)
biggest_element = 0
print '(10f5.2)',array
```

Output

```
[pointerf] arpointf:

 1.10 2.20 3.30 4.40
    5.50 9.90 8.80
    7.70 6.60 0.00
 Biggest element is
     9.89999962
 checking pointerhood
    : T
 1.10 2.20 3.30 4.40
    5.50 0.00 8.80
    7.70 6.60 0.00
```

You can base this off the file `arpointf.F90` in the repository

9. Linked list

- Linear data structure
- more flexible than array for insertion / deletion
- . . . but slower in access

Linked list



Insertion


Current
Next / Tail
Info
Next
Info
Next
Null
Info
Next
Null


Current
Next / Tail
Info
Next
Info
Next
Null
Info
Next
Null

10. Linked list datatypes

- Node: value field, and pointer to next node.
- List: pointer to head node.

```
type node
   integer :: value
   type(node),pointer :: next
end type node

type list
   type(node),pointer :: head
end type list
```

11. Sample main

Our main program will create three nodes, and attach them:

```
type(list) :: the_list
type(node),pointer :: node_ptr

nullify(the_list%head)

allocate(node_ptr); node_ptr%value = 1
call attach(the_list,node_ptr)
allocate(node_ptr); node_ptr%value = 5
call attach(the_list,node_ptr)
allocate(node_ptr); node_ptr%value = 3
call attach(the_list,node_ptr)

call print(the_list)
```

12. List initialization

```
subroutine attach( the_list,new_node )
  implicit none
  ! parameters
  type(list),intent(inout) :: the_list
  type(node),intent(inout),pointer :: new_node
```

First element becomes the list head:

```
! if the list has no head node, attached the new node
if (.not.associated(the_list%head)) then
   nullify(new_node%next)
   the_list%head => new_node
else
```

13. Attaching a node

New element attached at the end.

missing snippet listattacheadf

14. **Attaching a node**

Keep the list sorted: new largest element attached at the end.

```
allocate(new_node)
new_node%value = value
nullify(new_node%next)
current%next => new_node
```

Exercise 2

Write a `print` function for the linked list.
For the simplest solution, print each element on a line by itself.

More sophisticated: use the `Write` function and the `advance` keyword:

```
write(*,'(i1",")',advance="no") current%value
```

Exercise 3

Write a *length* function for the linked list.
Try it both with a loop, and recursively.